```
(*Coordinates for torus*)
M = 2;
x = {θ, φ};
X = {(a * Cos[θ] + b) Cos[φ], (a * Cos[θ] + b) Sin[φ], a * Sin[θ]};

In[11]:= (*Local coordinates*)
e = Table[D[X[[j]], x[[i]]], {i, 2}, {j, 3}] // ExpandAll // Simplify

Out[11]= {{-a Cos[φ] Sin[θ], -a Sin[θ] Sin[φ], a Cos[θ]},
  {-(b + a Cos[θ]) Sin[φ], (b + a Cos[θ]) Cos[φ], 0}}

In[12]:= (*Metric tensor*)
g = Table[e[[i]].e[[j]], {i, M}, {j, M}] // ExpandAll // Simplify

Out[12]= {{a^2, 0}, {0, (b + a Cos[θ])^2}}

In[13]:= Ig = Inverse[g] // ExpandAll // Simplify

Out[13]= {{1/a^2, 0}, {0, 1/(b + a Cos[θ])^2}}

In[14]:= (*Christoffel symbol of the first kind*)
Γ1 = Table[D[g[[i, j]], x[[k]]] + D[g[[i, k]], x[[j]]] - D[g[[j, k]], x[[i]]],
      {i, M}, {j, M}, {k, M}] / 2 // ExpandAll // Simplify

Out[14]= {{{0, 0}, {0, a (b + a Cos[θ]) Sin[θ]}},
  {{0, -a (b + a Cos[θ]) Sin[θ]}, {-a (b + a Cos[θ]) Sin[θ], 0}}}

In[15]:= (*Christoffel symbol of the second kind*)
Γ2 = Table[Sum[Ig[[i, j]] Γ1[[j, k, l]], {j, M}], {i, M}, {k, M}, {l, M}] // ExpandAll //
   Simplify

Out[15]= {{{0, 0}, {0, ((b + a Cos[θ]) Sin[θ])/a}}, {{0, -(a Sin[θ])/(b + a Cos[θ])}, {-(a Sin[θ])/(b + a Cos[θ]), 0}}}

In[16]:= (*Riemann curvature tensor*)
R = Table[D[Γ2[[i, j, k]], x[[l]]] - D[Γ2[[i, j, l]], x[[k]]] +
      Sum[Γ2[[m, j, k]] Γ2[[i, m, l]] - Γ2[[m, j, l]] Γ2[[i, m, k]], {m, M}],
     {i, M}, {j, M}, {k, M}, {l, M}] // ExpandAll // Simplify

Out[16]= {{{{0, 0}, {0, 0}}, {{0, -(Cos[θ] (b + a Cos[θ]))/a}, {Cos[θ] (b/a + Cos[θ]), 0}}},
  {{{0, (a Cos[θ])/(b + a Cos[θ])}, {-(a Cos[θ])/(b + a Cos[θ]), 0}}, {{0, 0}, {0, 0}}}}

In[17]:= cR = Table[Sum[g[[i, j]] R[[j, k, l, m]], {j, M}], {i, M}, {k, M}, {l, M}, {m, M}]

Out[17]= {{{{0, 0}, {0, 0}}, {{0, -a Cos[θ] (b + a Cos[θ])}, {a^2 Cos[θ] (b/a + Cos[θ]), 0}}},
  {{{0, a Cos[θ] (b + a Cos[θ])}, {-a Cos[θ] (b + a Cos[θ]), 0}}, {{0, 0}, {0, 0}}}}
```

In[18]:=
```
(*Ricci tensor*)
Ric = Table[Sum[R[[i, j, k, i]], {i, M}], {j, M}, {k, M}] // ExpandAll // Simplify
```

Out[18]=
$$\left\{\left\{\frac{a \operatorname{Cos}[\theta]}{b + a \operatorname{Cos}[\theta]}, 0\right\}, \left\{0, \operatorname{Cos}[\theta] \left(\frac{b}{a} + \operatorname{Cos}[\theta]\right)\right\}\right\}$$

In[19]:=
```
(*Scalar curvature*)
r = Sum[Ig[[i, j]] Ric[[i, j]], {i, M}, {j, M}] // ExpandAll // Simplify
```

Out[19]=
$$\frac{2 \operatorname{Cos}[\theta]}{a\, b + a^2 \operatorname{Cos}[\theta]}$$

In[20]:=
```
(*Covariant derivative of Riemann curvature tensor*)
DR = Table[D[cR[[i, j, k, l]], x[[m]]] - Sum[Γ2[[n, m, i]] cR[[n, j, k, l]], {n, M}] -
     Sum[Γ2[[n, m, j]] cR[[i, n, k, l]], {n, M}] - Sum[Γ2[[n, m, k]] cR[[i, j, n, l]],
      {n, M}] - Sum[Γ2[[n, m, l]] cR[[i, j, k, n]], {n, M}],
    {i, M}, {j, M}, {k, M}, {l, M}, {m, M}] // ExpandAll // Simplify
```

Out[20]=
$$\{\{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}\},$$
$$\{\{\{0, 0\}, \{a\, b \operatorname{Sin}[\theta], 0\}\}, \{\{-a\, b \operatorname{Sin}[\theta], 0\}, \{0, 0\}\}\}\},$$
$$\{\{\{\{0, 0\}, \{-a\, b \operatorname{Sin}[\theta], 0\}\}, \{\{a\, b \operatorname{Sin}[\theta], 0\}, \{0, 0\}\}\},$$
$$\{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}\}\}\}$$

In[22]:=
```
(*Bianchi identities*)
Table[DR[[i, j, k, l, m]] + DR[[i, j, l, m, k]] + DR[[i, j, m, k, l]], {i, M},
   {j, M}, {k, M}, {l, M}, {m, M}] // ExpandAll // Simplify
```

Out[22]=
$$\{\{\{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}\}, \{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}\}\},$$
$$\{\{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}\}, \{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}\}\}\}$$